
	Reason for change:
	1) As described in change 2.3, IMS PTC and UTRAN PTC needs to be synchronised for execution of procedure for removal of early dialog 

2) After aSRVCC HO, UE will have an early dialog. As per A.17.2 of 24.237, network will terminate early dialog of MT Call by sending ‘CANCEL’ for which UE  sends ‘487 terminated’ and ‘200 OK’ message. Now, SS responsds back ACK message.

 Note: A draft Prose CR would be raised for next RAN5 meeting on 36.523-1 and 34.229-1. In 34.229-1, Annex C.35 is defined to delete early dialog for MT Call.

To implement above behaviour, UTRAN PTC will send the co-oridnation signal to IMS PTC after SS sends CONNECT ACKNOWLEGE.(see change 2)

3) IP Routing table needs to be updated based dedicatedCell not based on cellID

4)  As described in change 1.2, UTRAN PTC and IMS PTC needs to be synchrozied during the deletion of early dialog

	
	

	Summary of change:
	1) Sent the co-ordination signal to EUTRA PTC

2) Received the co-ordination signal from UTRAN PTC,  SS sent CANCEL message and received 487 terminated and 200 OK from UE. And, SS responds with ACK message

3) Updated the IP routing with dedicatedCell using f_UTRAN_IP_Handling_Change_Configuration
4)  Synchrosed using f_IMS_IPCAN_SendCoOrdMsg(IMS and f_IMS_IPCAN_WaitForTrigger

Change 1
	Test case
	13.4.3.16

	Reason for change
	1) As described in change 2.3, IMS PTC and UTRAN PTC needs to be synchronised for execution of procedure for removal of early dialog 

2)  After aSRVCC HO, UE will have an early dialog. As per A.17.2 of 24.237, network will terminate early dialog of MT Call by sending ‘CANCEL’ for which UE  sends ‘487 terminated’ and ‘200 OK’ message. Now, SS responsds back ACK message.
Note: draft Prose CR would be raised for next RAN5 meeting on 36.523-1 and 34.229-1. In 34.229-1, Annex C.35 is defined to delete early dialog for MT Call.
To implement above behaviour, UTRAN PTC will send the co-oridnation signal to IMS PTC after SS sends CONNECT ACKNOWLEGE.(see change 2)
3) Need to wait for IMS deregistration from UTRAN PTC

	Summary of change
	1) Receive co-orination signal and sent  co-ordination after removal of early dialog

2) Received the co-ordination signal from UTRAN PTC,  SS sent CANCEL message and received 487 terminated and 200 OK from UE. And, SS responds with ACK message
3) Wait for co-ordination signal using f_IMS_IPCAN_WaitForTrigger()



	Source of change
	LTE_A\13\ASRVCC_EUTRA_IMS.ttcn


Before change:

	function f_TC_13_4_3_16_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp; /* @sic R5w140112: v_InviteRequestWithSdp instead of v_InviteRequest sic@ */

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionInProgress183;

    f_IMS_PTC_Init();

    f_IMS_Registration();

    // Steps 1 to 22 of the generic test procedure for IMS MT speech call (TS 36.508 4.5A.7.3-1).

    /* Where TS 36.508 4.5A.7.3-1 Steps 16 to 22 specifies:

       Steps 5-11 expected sequence defined in annex C.11 of TS 34.229-1 [35]. MTSI MT speech.

    */

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE();

    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);

    v_SessionInProgress183 := v_ResponseWithSdp.Response;

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_IMS_MTCallSetup_AnnexC11_Steps5_11(v_InviteRequestWithSdp, v_SessionInProgress183);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(OtherIPCAN); // @sic R5-142919 sic@

  }


After change :

	function f_TC_13_4_3_16_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp; /* @sic R5w140112: v_InviteRequestWithSdp instead of v_InviteRequest sic@ */

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionInProgress183;

    //134316 
    var InternetProtocol_Type v_Protocol;

    var template (value) MessageHeader v_MessageHeader_Cancel;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var SipUrl v_ContactUrl; 

    f_IMS_PTC_Init();

    f_IMS_Registration();

    // Steps 1 to 22 of the generic test procedure for IMS MT speech call (TS 36.508 4.5A.7.3-1).

    /* Where TS 36.508 4.5A.7.3-1 Steps 16 to 22 specifies:

       Steps 5-11 expected sequence defined in annex C.11 of TS 34.229-1 [35]. MTSI MT speech.

    */

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE();

    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);

    v_SessionInProgress183 := v_ResponseWithSdp.Response;

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_IMS_MTCallSetup_AnnexC11_Steps5_11(v_InviteRequestWithSdp, v_SessionInProgress183);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader); 

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

     //134316 NK

    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol(); //134314 NK UE changed the protocol in the PRACK as a response for '180 ringing'

                                                       //Setting the via header also same protocol as used to send previous SIP message

    v_MessageHeader_Cancel := f_IMS_CancelRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite); 

    IMS_Client.send(cas_IMS_Cancel_Request(cs_IMS_RoutingInfo(v_Protocol), cs_CANCEL_Request(v_ContactUrl, v_MessageHeader_Cancel)));

    interleave { // @sic R5-125773 sic@

      // Step 13 - The UE acknowledges the CANCEL with 200 OK.

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Cancel)))) {}

      // Step 14.  The UE responds to INVITE with a 487 Request Terminated final response after transaction was terminated.

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol),  cr_Response(c_statusLine487, f_IMS_InviteResponse_487_MessageHeaderRX(v_InviteRequestWithSdp.Invite.msgHeader)))) {}            //Anritsu TTCN CR R5s140178 MCC

    }

    // Step 15. Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.

    // FFS - Is this ok? Invite was cancelled

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite, c_statusLine487);   /* @sic R5s140092 change 1.3 sic@ */

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);
f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);
    f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(OtherIPCAN); // @sic R5-142919 sic@

  }


Change 2
	Test case
	13.4.3.16

	Reason for change
	1) IP Routing table needs to be updated based dedicatedCell not based on cellID

2) CONNECT need to be sent from SS with TI flag ‘0’ and CONNECT ACKNOWLEDGE need to be received with TI flag ‘1’

Note: A draft prose CR would be raised for the next RAN5 meeting

3) As described in change 1.2, UTRAN PTC and IMS PTC needs to be synchrosied during the deletion of early dialog
4) Need to send the coordintation for IMS deregistration

	Summary of change
	1) Updated the IP routing with dedicatedCell using f_UTRAN_IP_Handling_Change_Configuration
2) CONNECT sent with TI flag with ‘0’ and CONNECT ACKNOWLEDGE received with flag ‘1’

3)  Synchrosed using f_IMS_IPCAN_SendCoOrdMsg(IMS and f_IMS_IPCAN_WaitForTrigger
4) Sent coordination for using f_IMS_IPCAN_SendCoOrdMsg()

	Source of change
	LTE_A\13\ASRVCC_EUTRA_UTRAN.ttcn


	function f_TC_13_4_3_16_UTRAN() runs on UTRAN_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 34.108 cl. 6.10.2.4.1.26

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 26" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5); // ToDo Cs domain

    f_UTRAN_IP_Handling_Change_Configuration (utran_Cell5,  cell_DCH_64kPS_RAB_SRB);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    // @siclog "Steps 32 to 37" siclog@

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, -, true);

    f_UT_AnswerCall ( UT );

    v_TI_R.tiVal  := '000'B;

    v_TI_R.tiFlag := '0'B;
    //@siclog "Step 38" siclog@

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Connect(v_TI_R)));

    v_TI_S := v_TI_R;

    v_TI_S.tiFlag := '1'B;
    //@siclog "Step 39" siclog@

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_ConnectAck(v_TI_S)));

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }

  


After change :

	function f_TC_13_4_3_16_UTRAN() runs on UTRAN_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 34.108 cl. 6.10.2.4.1.26

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 26" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5); // ToDo Cs domain

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    // @siclog "Steps 32 to 37" siclog@

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, -, true);

    f_UT_AnswerCall ( UT );

    v_TI_R.tiVal  := '000'B;

    v_TI_R.tiFlag := '1'B; //134316
    //@siclog "Step 38" siclog@

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Connect(v_TI_R)));

    v_TI_S := v_TI_R;

    v_TI_S.tiFlag := '0'B; //134316 

    //@siclog "Step 39" siclog@

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_ConnectAck(v_TI_S)));

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // IMS deregistration
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }

  


Change 3-8
Change5 to 10 of R5s140572 are required for this testcase

